
Beispiel: Klimasteuerung

Am Beispiel eine einfachen Klimasteuerung wird gezeigt, dass der Commodore 64 mehr als
ein Spielecomputer ist. Er ist mit wenig Zubehör als Steuerungsrechner einsetzbar. Ferner ist
es durchaus möglich, mit dem eingebauten BASIC strukturierte Programme zu schreiben.

Aufbau der Hardware
Die verwendete Hardware wurde in den 1980er Jahren von Hr. Schill an der Volkshochschule
Darmstadt im Rahmen des Kurses „Messen – Steuern – Regeln mit dem C64“ vorgestellt und
vertrieben.

• Am Control-Port 1 (Joystick) ist mittels einer Zusatzplatine ein 100 k ¿ NTC
(temperaturabhängiger Widerstand) angeschlossen.

• Am User-Port ist eine 4fach-Relaiskarte angeschlossen, wobei die einzelnen Relais
folgende Bedeutung haben:
• Relais 1: kühlen
• Relais 2: heizen
• Relais 3: Luft umwälzen (Temperatur o.k.)

Ansicht Relaiskarte:

Ansicht Controlport-Karte mit temperaturabhängigem Widerstand:

Software
Das hier gezeigte Beispielprogramm ist wie folgt strukturiert:

ab 2000: DATAs zur Umrechnung des digitalisierten Wertes in Temperatur (Grad C)
ab 5000: Unterprogramm zur Temperaturmessung mit AD-Wandler 1
ab 5100: Unterprogramm zur Temperaturmessung mit AD-Wandler 2
ab 6000: Unterprogramm: alle 4 Relais einschalten
ab 6100: Unterprogramm: Relais 1 einschalten
ab 6200: Unterprogramm: Relais 2 einschalten
ab 6300: Unterprogramm: Relais 3 einschalten
ab 6400: Unterprogramm: Relais 4 einschalten
ab 7000: Unterprogramm: alle 4 Relais ausschalten
ab 7100: Unterprogramm: Relais 1 ausschalten
ab 7200: Unterprogramm: Relais 2 ausschalten
ab 7300: Unterprogramm: Relais 3 ausschalten
ab 7400: Unterprogramm: Relais 4 ausschalten

1 REM EINFACHE HEIZUNGSSTEUERUNG V5
2 REM 2003-04-18 MS  NEU
3 REM 2003-05-01 MS  ZWEI SCHWELLWERTE
4 REM
10 REM TEMPERATURTABELLE BAUEN
15 DIM GRAD(255)
16 FOR I=0 TO 255
17 READ GRAD(I)
18 NEXT  I
20 REM *********************
21 REM KONSTANTEN DEFINIEREN
22 REM *********************
30 A1 = 54297 : REM AD-WANDLER 1
40 A2 = 54298 : REM AD-WANDLER 2
50 DR = 56577 : REM DATENREGISTER
60 RR = 56579 : REM RICHTUNGSREGISTER
100 REM *********
101 REM VARIABLEN
102 REM *********
110 DIM RELAIS(4) : REM RELAIS 1-4
111 FOR I=1 TO 4 : RELAIS(I)=0 : NEXT I
120 D1 = 0 : REM DIGITALWERT FUEHLER 1
130 T1 = 0 : REM TEMP.WERT FUEHLER 1
140 D2 = 0 : REM DIGITALWERT FUEHLER 2
150 T2 = 0 : REM TEMP.WERT FUEHLER 2
160 SU = 0 : REM UNTERER SCHWELLWERT
170 SO = 0 : REM OBERER SCHWELLWERT
180 ZY = 0 : REM ANZAHL ZYKLEN
999 REM
1000 REM *************
1001 REM HAUPTPROGRAMM
1002 REM *************
1003 REM
1010 GOSUB 7000 : REM ALLE RELAIS AUS
1020 PRINT "UNTERER SCHWELLWERT "
1030 INPUT  SU
1040 PRINT "OBERER SCHWELLWERT "
1050 INPUT  SO
1060 IF SU>SO GOTO 1010
1070 PRINT "ZYKLEN"
1080 INPUT ZY
1090 GOSUB 7000 : REM ALLE RELAIS AUS
1100 FOR Z=1 TO ZY
1110 GOSUB 5100 : REM MESSEN
1120 IF T2 < SU GOTO 1500 : REM ZU KALT
1130 IF T2 > SO GOTO 1600 : REM ZU WARM
1140 GOTO  1700 : REM TEMPERATUR OK
1150 NEXT Z
1160 END
1500 REM ZU KALT

1505 PRINT "ZYKLUS:";Z;"TEMP:";T2;" --> ZU KALT"
1510 GOSUB 7100 : REM KUEHLUNG AUS
1520 GOSUB 7300 : REM VENTILATOR AUS
1530 GOSUB 6200 : REM HEIZUNG AN
1540 FORX=1TO500:NEXTX
1550 GOTO 1150
1600 REM ZU WARM
1605 PRINT "ZYKLUS:";Z;"TEMP:";T2;" --> ZU WARM"
1610 GOSUB 7200 : REM HEIZUNG AUS
1620 GOSUB 7300 : REM VENTILATOR AUS
1630 GOSUB 6100 : REM KUEHLUNG AN
1640 FORX=1TO500:NEXTX
1650 GOTO 1150
1700 REM TEMP OK
1705 PRINT "ZYKLUS:";Z;"TEMP:";T2;" --> TEMP. OK"
1710 GOSUB 7100 : REM KUEHLUNG AUS
1720 GOSUB 7200 : REM HEIZUNG AUS
1730 GOSUB 6300 : REM VENTILATOR AN
1740 FORX=1TO500:NEXTX
1750 GOTO 1150
1999 END
2000 REM DATAS FUER 100KOHM FUEHLER
2005 DATA97,96,95,93,91,90,88,85,83,80,78,75,73,72,71,70,68,66,65,64,62,60,59,58
2010 DATA56,55,54,53,52,51,51,50.5,50,49,48.5,48,47.5,47,46.5,46,45.5,45,44.5,44
2020 DATA43.5,43,42.5,42,41.5,41,40.7,40.4,40,39.5,39,38.7,38.4,38,37.5,37,36.5,36,35
2030 DATA34.7,34.4,34,33.8,33.6,33.4,33.2,33,32.8,32.6,32.4,32.2,32,31.5,31,30,30
2040 DATA29.5,29.5,29,28.5,28.5,28,28,28,27.5,27.5,27,27,26.5,26.5,26
2050 DATA26,25.5,25.5,25,25,24.5,24.5,24.5,24,24,24,23.5,23.5,23.5
2060 DATA23,23,23,22.5,22.5,22.5,22,22,22
2070 DATA21.5,21,21,21,20.5,20.5,20.5,20,20,20,19.5,19.5,19.5,19,19,19,18.5,18.5,18.5
2080 DATA18,18,18,17.5,17.5,17.5,16,16,16,15.5,15.5,15.5,15,15,15,14.5,14.5,14.5
2090 DATA14,14,14,14,13.5,13.5,13.5,13,13,13,13,12.5,12.5,12.5,12,12,12,12,11.5,11.5
3000 DATA11.5,11,11,11,11,10.5,10.5,10.5,10,10,10,9.5,9.5,9.5,9,9,9,9,9,9,9,9,9,9,9,9
3010 DATA 8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8,8
3020 DATA8,8,8,8,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7,7
5000 REM SUBROUTINE TEMP-FUEHLER 1
5010 D1 = PEEK(A1)
5020 T1 = GRAD(D1)
5030 RETURN
5100 REM SUBROUTINE TEMP-FUEHLER 2
5110 D2 = PEEK(A2)
5120 T2 = GRAD(D2)
5130 RETURN
6000 REM SUBROUTINE ALLE RELAIS AN
6010 FOR I=1 TO 4
6015 RELAIS(I) = 1
6020 NEXTI
6030 A=0
6040 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
6050 POKE RR,255 : REM AUSGANG
6060 POKE DR,A
6070 RETURN
6100 REM SUBROUTINE RELAIS 1 AN
6110 RELAIS(1) = 1
6120 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
6150 POKE RR,255 : REM AUSGANG
6160 POKE DR,A
6170 RETURN
6200 REM SUBROUTINE RELAIS 2 AN
6210 RELAIS(2) = 1
6220 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
6250 POKE RR,255 : REM AUSGANG
6260 POKE DR,A
6270 RETURN
6300 REM SUBROUTINE RELAIS 3 AN
6310 RELAIS(3) = 1
6320 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
6350 POKE RR,255 : REM AUSGANG
6360 POKE DR,A
6370 RETURN
6400 REM SUBROUTINE RELAIS 4 AN
6410 RELAIS(4) = 1
6420 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
6450 POKE RR,255 : REM AUSGANG
6460 POKE DR,A
6470 RETURN
7000 REM SUBROUTINE ALLE RELAIS AUS

7010 FOR I=1 TO 4
7015 RELAIS(I) = 0
7020 NEXTI
7030 A=0
7040 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
7050 POKE RR,255 : REM AUSGANG
7060 POKE DR,A
7070 RETURN
7100 REM SUBROUTINE RELAIS 1 AUS
7110 RELAIS(1) = 0
7120 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
7150 POKE RR,255 : REM AUSGANG
7160 POKE DR,A
7170 RETURN
7200 REM SUBROUTINE RELAIS 2 AUS
7210 RELAIS(2) = 0
7220 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
7250 POKE RR,255 : REM AUSGANG
7260 POKE DR,A
7270 RETURN
7300 REM SUBROUTINE RELAIS 3 AUS
7310 RELAIS(3) = 0
7320 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
7350 POKE RR,255 : REM AUSGANG
7360 POKE DR,A
7370 RETURN
7400 REM SUBROUTINE RELAIS 4 AUS
7410 RELAIS(4) = 0
7420 A=RELAIS(1)*1 OR RELAIS(2)*2 OR RELAIS(3)*4 OR RELAIS(4)*8
7450 POKE RR,255 : REM AUSGANG
7460 POKE DR,A
7470 RETURN


